
数据处理是对数据的采集、存储、检索、加工、变换和传输的过程。

数据处理的基本目的是从大量的、可能是杂乱无章的、难以理解的数据中抽取并推导出对

我们来说是有价值、有意义的数据。

举个最简单的例子，家里各种各样的东西一大堆扔在一个角落，这些东西因为没有区分显

得非常乱，看不出哪些是有用的，哪些是没用的。如果抽出时间，将这些东西整理区分，重新

摆放。在整洁的环境中，则可以快速地找到对你有价值的东西。家里的东西乱了要整理，数据

乱了也要整理，经过整理，留下对我们最有用的数据，其他都剔除。

6.1　如何将不规范数据源
转换成规范数据源

很多时候，数据并不仅仅存在于Excel中，有可能来自网站、数据库、文本……需要将数

据导入到Excel中，然后进一步处理才可以使用。

6.1.1　将记事本的数据导入Excel

卢子：以前的数据都是直接在Excel中输入的。如图6-1所示，如果是在记事本中输入的，你懂得如何导入

到Excel中吗？

木木：这个太简单了。

STEP 01  如图6-2所示，打开记事本，选择里面所有的内容，按组合键Ctrl+C。

图 6-1　在记事本中输入的数据 图 6-2　复制内容

STEP 02  如图6-3所示，打开Excel，选中单元格A1，按组合键Ctrl+V就搞定了。

图 6-3　粘贴内容

卢子：现在这些数据是在同一个单元格，你知道怎么分开吗？

木木：这个借助分列功能就可以分开吧。

STEP 01  如图6-4所示，选择区域A1:A15，切换到“数据”选项卡，单击“分列”按
钮，弹出“文本分列向导”对话框，选中“分隔符号”单选按钮，单击“下一步”按钮。

STEP 02  如图6-5所示，选中“其他”复选框，输入中文状态下的逗号（，），单击

“完成”按钮。

图 6-4　文本分列向导第 1 步

图 6-5　文本分列向导第 2 步

怎么回事呢？如图6-6所示，员工编号前面的0消失了，“帐号”列中的内容变成6.26346E+16。

图 6-6　数字出现异常

卢子：因为“员工编号”与“帐号”两列都需要将单元格设置为文本格式才行，否则就出错。

STEP 01  重复刚才的操作，在“文本分列向导”第2步设置后，单击“下一步”按
钮，进入第3步，选中“员工编号”这一列，然后选中“文本”单选按钮，如图6-7所示。

图 6-7　将员工编号设置为文本格式

STEP 02  如图6-8所示，选中“帐号”这一列，然后选中“文本”单选按钮，单击“完
成”按钮。

图 6-8　将账号设置为文本格式

通过这小小的改变，如图6-9所示，员工编号与账号就恢复正常了。

图 6-9　导入后数据正常了

木木：原来分列还有这样的功能，学习了。

6.1.2　去除错误值

卢子：这样虽然将数据分开，但里面还是包含了一些错误值“#N/A”，这些看起来很不美观。如何去除

这些错误值呢？

木木：哈哈，别的不会，替换这一招我用得非常熟练。

STEP 01  选择任意错误值的单元格后，复制。

STEP 02  如图6-10所示，按组合键Ctrl+H打开“查找和替换”对话框，将错误值粘贴
在“查找内容”下拉列表框中，然后单击“全部替换”按钮。

图 6-10　将错误值替换为空

如图6-11所示，替换后，错误值就全部变成空白了。

图 6-11　替换后的效果

卢子：木木好样的，查找和替换以后对你而言，再无难题！实际上替换掉这些错误值后还得重新确认，

补充正确的信息，这样得到的才是完整的信息。

6.2　如何用逻辑函数
进行条件判断

你是广东的吗？

你是学财务的吗？

你是Excel爱好者吗？

……

每天都会接触到很多类似的问题，都围绕着“是”或者“不是”展开。“是”在Excel中
用TRUE表示，“不是”在Excel中用FALSE。而TRUE与FALSE就是逻辑函数，也就是说，我

们每天都在跟逻辑函数打交道。

6.2.1　IF函数判断称呼

卢子：如图6-12所示，这是一份学生成绩表，如何根据性别判断称呼，性别为男的显示“先生”，性别

为女的显示“女士”？

木木：条件判断不就是IF函数吗，很简单。

STEP 01  如图6-13所示，选中D2单元格，在编辑栏中输入下面的公式。

=IF(C2="男","先生","女士")

图 6-12　学生成绩表

图 6-13　输入 IF 函数

STEP 02  如图6-14所示，按Enter键后，D2单元格自动生成“先生”。把鼠标指针放在
D2单元格的右下角，出现“+”时，双击单元格。

如图6-15所示，填充公式后，所有称呼都显示出来。

卢子：不错，我再补充一下用法，你就当复习，温故而知新。

如图6-16所示，IF函数有三个参数，每个参数都有不同的含义，只有明白了其中的含义，

才能准确地设置公式。

图 6-14　双击填充公式

图 6-15　填充公式后的效果

图 6-16　IF 函数语法

刚刚的判断也可以改成下面的公式。

=IF(C2="女","女士","先生")

木木：再复习几次，我都可以当老师了，哈哈！

6.2.2　IF函数嵌套判断专业代号

卢子：刚刚性别只有两种情况，非男即女。现在专业代号有三种，“理工”显示LG，“文科”显示

WK，“财经”显示CJ。单个IF函数是无法直接完成的，你知道怎么做吗？

木木：函数嵌套我还不会，教教我怎么做吧。

卢子：函数最有意思的地方就是嵌套，每个参数都可以嵌套不同的函数，从而变成非常强大的公式。与

组合积木差不多，通过小小的积木，组合成庞大的模型。

=IF(E2="理工","LG",IF(E2="文科","WK","CJ"))

如图6-17所示，当E2单元格中是“理工”时显示LG，否则就显示后面的IF(E2="文科", 
"WK","CJ")。

图 6-17　IF 函数分步解读

执行了第一次判断后，再执行第二次判断。

当E2是“文科”时显示WK，否则就显

示CJ。
木木：听起来还是有点模糊。

卢子：我再用一个示意图来表示，你一看就懂。

如图6-18所示，其实IF函数就跟找女朋友一

样，首先是判断美丑，如果是美女再进一

步判断是否聊得来。

图 6-18　找女友示意图

木木：原来你们男人都是这样，看脸的。

卢子：其实女人也差不多，经常都听见女人说这

么一句：你是个好人，如图6-19所示。

图 6-19　好人图

木木：哈哈，没错，卢子，你是个好人！

卢子：每次听到这句话都有一种欲哭无泪的感

觉。不说这个了，继续回到 IF函数的运 

用上。

6.2.3　IF函数嵌套的巩固

卢子：利用前面的知识，获取总分。来源为“本地”，总分为原始分加30；来源为“本省”，总分为原

始分加20；来源为“外省”，总分为原始分加10。
木木：我试试，这个我应该会做。

STEP 01  在I2单元格中输入公式。

=IF(G2="本地",H2+30,IF(G2="本省",H2+20,H2+10))

STEP 02  把鼠标指针放在I2单元格右下角，出现“+”时，双击单元格，填充公式。

依样画葫芦，搞定！

卢子：写得不错，但这个公式还可以进一步简

化。这里就涉及数学中的合并同类项，就

是将相同的内容提取出来，对表达式进行

简化，如图6-20所示。

图 6-20　合并同类项

其实Excel中的公式跟数学中的表达式也有点类似，可以做同样的操作。“H2+”这个是同

样的，所以可以提取出来，最终公式为：

=H2+IF(G2="本地", 30,IF(G2="本省", 20,10))

木木：原来这样，那我数学不好是不是不能学好公式？

卢子：数学好对学好公式有一点作用，但也不是绝对的。再说，实际工作中只要能解决问题就行，不要

执着于公式的简化。简化公式这个只在学习的过程中可以用，实际上不提倡！

木木：这样还好，要不然我都没信心了。

6.2.4　AND函数满足多条件获取录取情况

卢子：截至目前都是单个IF函数的运用，现在开始会涉及与其他函数的嵌套运用。

木木：函数嵌套这个一直是我的心结，单个函数我还懂，一嵌套就晕了。

卢子：其实只要能熟练单个函数的用法，多个函数的嵌套也不是难事。现在给你介绍满足多条件获取录

取的情况。

现在某公司准备录取性别为女性，总分在600分以上的人，该怎么做呢？

在J2单元格中输入公式，并向下填充公式。

=IF(AND(C2="女",I2>600),"录取","")

如图6-21所示，AND函数当所有条件都为TRUE时，返回TRUE。
如图6-22所示，AND函数只要其中一个条件为FALSE，则返回FALSE。

图 6-21　AND 函数语法条件（1）
图 6-22　AND 函数语法条件（2）

举一个简单的例子来说明一下，怎么算在谈恋爱呢？

条件1：男的喜欢女的。

条件2：女的喜欢男的。

只有同时满足这两个条件，才算谈恋爱，否则最多算单相思。

=IF(AND("男的喜欢女的","女的喜欢男的"),"谈恋爱","单相思")

木木：秒懂！卢子现在谈恋爱了，说话变得越来越有才，哈哈哈。

卢子：其实很多事情都是相通的，你想学习Excel，我愿意分享Excel，才有了这次对话。

说到AND函数不得不提另外一个函数：OR，这个函数跟AND函数很相似。

如图6-23所示，OR函数只有当所有条件都为FALSE的时候，才返回FALSE。
如图6-24所示，OR函数只要其中一个条件为TRUE，则返回TRUE。

图 6-23　OR 函数语法条件（1）
图 6-24　OR 函数语法条件（2）

举个例子来说明下，怎样才算好运？

条件1：出门捡到钱了。

条件2：买彩票中奖了。

条件3：遇到好心人帮你解决疑难了。

我们不需要所有条件都成立才算好运，只需满足其中一个条件即可。

= I F ( O R ( " 出门捡到钱了 " ,  " 买彩票中奖了 " ,  " 遇到好心人帮你解决疑难了 " ) , " 好运 " , 
"正常")

木木：我好幸运啊，遇到卢子大帅哥教我Excel，解决疑难。

展知 识 扩
AND函数可以用“*”代替，OR函数可以用“+”代替。

判断A1单元格是否在区间0～60之间，满足显示“不及格”，不满足显示“及格”。

=IF(AND(A1>0,A1<60),"不及格","及格")
=IF((A1>0)*(A1<60),"不及格","及格")

判断A1单元格是否小于160或者大于170，满足显示“不合格”，不满足显示“合格”。

=IF(OR(A1<160,A1>170),"不合格","合格")
=IF((A1<160)+(A1>170),"不合格","合格")

6.3　如何用数学与统计函
数进行数据汇总

总分考了多少？

最高分是多少？

最低分是多少？

全班有多少人？

……

数学与统计函数同样跟我们的生活息息相关，非常重要。

6.3.1　自动求和妙用

卢子：如图6-25所示，这是某学校的成绩明细表，如何统计总分、平均分、考试人数、最高分、最低分？

图 6-25　成绩明细表

木木：总分这个我会，其他我就不懂了。

STEP 01  如图6-26
所示，选中E2单元格，切

换到“公式”选项卡，单

击“自动求和”按钮，就

自动帮你选择区域求和。

图 6-26　自动求和

STEP 02  如图6-27所示，将公式下拉填充到E12，搞定。

图 6-27　填充公式

卢子：“自动求和”这个功能确实很实用，轻轻一点就全搞定。其实“自动求和”并不仅仅是求和而

已，还包含了很多功能。如图6-28所示，单击“自动求和”下拉按钮，会出现“求和”“平均

值”“计数”“最大值”“最小值”等命令。

图 6-28　显示自动求和隐藏的功能

木木：天啊，居然藏着这么多秘密！

卢子：这个也是我无意间发现的，那时无聊，就对着Excel各个功能乱点，点到这个时就像发现新大陆一

样。这几个你可以逐个去测试，我把公式先发给你看看。最终效果如图6-29所示。

平均分：

=AVERAGE(B2:D2)

考试人数：

=COUNT(B2:B12)

最高分：

=MAX(B2:B12)

最低分：

=MIN(B2:B12)

图 6-29　效果图

需要注意的是，区域记得更改，智能选择的区域不一定正确，这几个函数都比较简单，会

一个其他也就都会了。

木木：是啊，一下子5个函数都学会了，我好厉害啊！

展知 识 扩

将鼠标指针移到“自动

求和”按钮上面，就会出现

求和的快捷键为Alt+=，如 
图6-30所示。

图 6-30　求和的快捷键

这个快捷键非常有用，如图6-31所示，对行列求总分。只需选择整个区域，按快捷键

Alt+=，瞬间就完成了行列总分。

图 6-31　快捷键的运用

6.3.2　SUMIF（COUNTIF）函数对科目进行单条件求和与计数

卢子：前面5个函数都比较简单，直接用“自动求和”下拉按钮就可以搞定，不伤脑。下面这两个问题就

稍微有点难度！

如图6-32所示，根据左边每个科目的消费明细，统计右边的科目出现的次数与金额。

图 6-32　每个科目的消费明细

如图6-33所示，我们知道COUNT函数是

计数，IF函数是条件，两个合起来就是条件

计数。

图 6-33　COUNT 与 IF 函数合并图

统计科目划分的次数就可以用：

=COUNTIF(B:B,E2)

在F2单元格中输入公式，并填充公式到

F5单元格中。

如图6-34所示，再来看看这个函数的语法。

木木：原来函数可以这么玩啊，长见识了！如 

图6-35所示，那按条件统计金额不就可以用

SUMIF函数。

图 6-34　COUNTIF 函数语法

图 6-35　SUM 与 IF 合并图

卢子：木木好聪明啊，举一反三。

木木：不过我不懂SUMIF函数的用法，你给我讲

讲吧。

卢子：如图6-36所示，SUMIF函数比COUNTIF函

数多一个求和区域而已，其他都一样。很

多人说函数难，那是因为找不到方法，如

果方法懂了，函数真的很简单，学会一

个，其他相关联的就都会了。

图 6-36　SUMIF 函数语法

木木：我来试试怎么写公式。

条件区域是B:B；
条件是E2；
求和区域是C:C；
综合起来就是：

=SUMIF(B:B,E2,C:C)

卢子：还有一个常用的函数AVERAGEIF，是按条

件求平均值，语法跟SUMIF函数一样。

=AVERAGEIF(B:B,E2,C:C)

但MAXIF跟MINIF Excel暂时不支持。

6.3.3　SUMIFS（COUNTIFS）函数对部门、科目进行多条件求和与计数

卢子：说完单条件，必须说多条件。如图6-37所示，对部门、科目两个条件，进行次数与金额统计。

图 6-37　对部门与科目多条件统计

木木：虽然我不懂，但我猜测应该是用COUNTIF与SUMIF函数再加点什么组成一个新函数完成。

卢子：猜的没错，英语中的复数很多都是直接在后面加s，表示多于一次，如sea-seas, girl-girls, day-days。

也就是说，多条件其实可以在后面加个S，如图6-38所示，COUNTIF-COUNTIFS，
SUMIF-SUMIFS。

图 6-38　COUNTIFS 函数语法

木木：原来语法跟COUNTIF函数一样，只是多几个条件区域和条件，我会用了。

在G2单元格中输入公式，并下拉填充公式。

=COUNTIFS(A:A,E2,B:B,F2)

现在发觉我没那么怕公式了，一学就会，我好聪明啊！

卢子：是啊，好厉害啊。

如图6-39所示，我再跟你说下SUMIFS函数的语法：

图 6-39　SUMIFS 函数语法

SUMIFS函数与COUNTIFS函数有点像，条件区域与条件是一一对应的，只是在第一参数

位置写求和区域。

木木：那我也会用了。

在H2单元格中输入公式，并下拉填充公式。

=SUMIFS(C:C,A:A,E2,B:B,F2)

最终效果，如图6-40所示。

图 6-40　最终效果图

卢子：如果所有人都像你这么聪明的话，那就好了。

6.3.4　SUMPRODUCT函数实现加权得分

卢子：如图6-41所示，年底自评，要对项目进行加

权得分，你知道怎么做吗？

图 6-41　对项目进行加权得分

木木：这个很简单。

STEP 01  在D2单元格中输入公式，
并下拉填充公式。

=B2*C2

STEP 02  在D9单元格中输入公式进
行求和，如图6-42所示。

=SUM(D2:D8)

卢子：这个也是一种办法，但其实Excel内置就有

这个函数，可以不用借助辅助列完成。如

图6-43所示，一起来看下SUMPRODUCT函

数的用法。

图 6-42　分步求和

图 6-43　SUMPRODUCT 函数语法

=SUMPRODUCT(B2:B8,C2:C8)

等同于：

=B2*C2+B3*C3+B4*C4+B5*C5+B6*C6+B7*C7+ 
B8*C8

如图6-44所示，其实SUMPRODUCT函数

同样是有两个函数组成，一个是SUM函数，

另一个是PRODUCT函数。

图 6-44　SUM 与 PRODUCT 合并图

PRODUCT函数就是乘积，比如，要计算B2单元格与C2单元格的乘积，就用：

=PRODUCT(B2:C2)

木木：原来如此，Excel的函数都是玩组合的，有点意思。

6.3.5　TRIMMEAN函数去除最大值与最小值后求平均值

卢子：如图6-45所示，我们经常会看到在比赛的时候，评委评分都会去除最大值和最小值，然后求平均

数，这个你懂得怎么操作吗？

图 6-45　评委评分

木木：这个结合前面的知识点我可以做出来，先用SUM函数求和，然后依次用MAX函数求最大值，用

MIN函数求最小值，用总和减去最大值与最小值，最后除以8就搞定。

=(SUM(B2:K2)-MAX(B2:K2)-MIN(B2:K2))/8

卢子：木木真的越来越牛了，什么问题都难不倒你，常规函数用得越来越熟！

这里跟你介绍一个不是很常用的TRIMMEAN函数，专门做这些去除异常值的事。

如图6-46所示，是TRIMMEAN函数语法说明。

极值比例我简单说明下，如果要去除最大值与最小值，就是去除20%，也就是0.2；如果要

去除前2大前2小，就是0.4。

图 6-46　TRIMMEAN 函数语法

也就是说去掉最大值和最小值后的平均值为：

=TRIMMEAN(B2:K2,0.2)

木木：原来如此简单，我还笨笨地用了那么多函数！

6.4　如何借助查找与引
用函数在万千数据
中获取对应值

时不时我们可以看到这一幕。

某人坐在电脑前，熟练按着键盘，Ctrl+C、Ctrl+F、Ctrl+V。仔细观察的话，你可以看

见他原来是在根据某个项目在另外一个表中查找相应的对应值，复制、查找、粘贴，如此 
循环。

STOP！Excel中提供了VLOOKUP等一系列查找与引用函数，分分钟帮你找到对应值，别

做这些无用功了。

6.4.1　VLOOKUP函数根据姓名查找职业

卢子：如图6-47所示，这里有一份人员信息对应表，如何通过姓名，查找对应的职业？

图 6-47　人员信息对应表

木木：这个我想到了两种方法。

① 复制姓名，然后用查找功能，找到对应值，粘贴上去；

② 复制姓名，然后用筛选功能，筛选出对应值，粘贴上去。

卢子：现在的姓名只有5个，用不了2分钟就搞定，但如果是500个、5000个姓名？

木木：那我就只有躲在墙角哭的份儿，这么多，加班加点的节奏。

卢子：这时就是VLOOKUP函数显神威的时刻，用这么一句话形容VLOOKUP函数：自从学了VLOOKUP

函数，腿也不疼了，腰也不酸了，吃嘛嘛香，身体倍棒。

木木：疗效这么好，我也想学一学！

卢子：这个函数有点难，有4个参数，我先慢慢跟你说。

如图6-48所示，VLOOKUP函数语法。

图 6-48　VLOOKUP 函数语法

如图6-49所示，根据实例来说明会更加清楚。

图 6-49　图解 VLOOKUP 函数

木木：看到你这个图，多看两遍，发觉我都能看懂了。

6.4.2　VLOOKUP函数根据姓名按顺序查找多列对应值

卢子：既然你都会了，那我就来考考你。如图6-50所示，如何根据姓名，依次返回性别、公司名称、职

业、学历？

图 6-50　多条件查询

木木：这个难不倒我。

查找的值：G2
要查找的区域：A:E
匹配方式：0（精确查找）

唯一不同的是，返回区域的第几列，分别是：2、3、4、5。
H2单元格的公式：

=VLOOKUP(G2,A:E,2,0)

I2单元格的公式：

=VLOOKUP(G2,A:E,3,0)

J2单元格的公式：

=VLOOKUP(G2,A:E,4,0)

K2单元格的公式：

=VLOOKUP(G2,A:E,5,0)

卢子：不错，这也是一种办法。因为VLOOKUP函数的其他三个参数都是固定的，只有一个是变动的，

这时也可以借助其他方法来完成。

要返回列号，其实可以借助COLUMN函数，这个函数非常简单，只有一个参数。

如图6-51所示，在任意单元格中输入公式，然后向右拖动，就可以自动生成1～N。

=COLUMN(A1)

图 6-51　借助 COLUMN 函数生成序号

如果细心的话，可以看到一个问题，就是里面的参数A1，在向右拖动公式的时候会变成

B1、C1、D1、E1，也就是不固定下来。同理VLOOKUP函数的第一个参数如果随着一起向右

拖动公式也会改变。

那怎么处理呢？

木木：这个好像用什么引用方式就可以？以前用过，现在不记得了。

卢子：如图6-52所示，输入公式后，

不要急着按Enter键。选择G2

单元格，然后按F4键，注意观

察编辑栏的变化，这时自动添

加了两个美元符号（$）。

图 6-52　F4 键的使用方法

如图6-53所示，通过不断按F4
键，会分别改变美元符号（$）的

位置。

图 6-53　切换 F4 键的效果图

这个美元符号（$）有什么作

用呢？如图6-54所示。

（1）相对引用：就是行列都

不给美元符号，这样公式复制到

哪里，哪里就跟着变。

（2）绝对引用：行列都给美

元符号，不管怎么复制公式，就

是不会变。

（3）混合引用：只给行或者

列美元符号。给行美元符号，行

不变；给列美元符号，列不变。

图 6-54　美元符号的作用

最终公式：

=VLOOKUP($G2,$A:$E,COLUMN(B1),0)

木木：原来Excel也爱钱，中国人就用人民币，美国人就用美元，给点美元就能收买美国人开发的Excel，
塞点美元，Excel全听你指挥。

6.4.3　VLOOKUP函数根据公司简称获取电话

卢子：如图6-55所示，在很多时候，我们输入公司名称都不会按全名输入，只是输入简称而已，如威航

货运有限公司，就输入威航货运，现在要如何根据简称获取电话呢？

图 6-55　根据公司简称获取电话

木木：原来都是这么懒，以为只有我一个人这么做。前面说过如果VLOOKUP函数第四参数设置为“1”
就是模糊查找，应该是利用

这个特点来完成的。

如图6-56所示，在E2单元格

中输入公式，并向下填充公式。

图 6-56　错误的查询方法

=VLOOKUP(D2,A:B,2,1)

怎么回事呢？怎么结果会这样子呢？

卢子：VLOOKUP函数的模糊匹配不是这么用的，而是运用在其他场合，等下再跟你说。这里涉及一个

新知识点，就是通配符的使用。

在Excel中有两种通配符，分别是星号（*）和问号（?）。

星号（*）代表所有字符。

问号（?）代表一个字符。

我举个例子说明一下，我的全名是3个字符，卢是最后一个字，这时可以这么表示：??卢。

如果我现在没有给你提示是多少个字符，也就是有可能是2个、3个或4个，这时就得用：*卢。

因为是让你猜全名，所以前面的字符都是不确定的，也就是会用到通配符。

木木：貌似懂了一点，你再说说这个具体如何使用？

卢子：回到实际例子，“威航货运”就是要查找威航货运有限公司的对应电话，就得用：“威航货运

*”，也就是说查找第一个可以用：

=VLOOKUP("威航货运*",A:B,2,0)

但总不能每个都改一下吧，这时就得利用一个文本连接符&，将单元格与星号（*）连接

起来。这个就像月老一样，给男女牵线，最后结合在一起！

="男"&"女"="男女"

综合起来就是：

=VLOOKUP(D2&"*",A:B,2,0)

木木：这样子啊，懂了。

6.4.4　VLOOKUP函数模糊匹配获取等级

卢子：现在来跟你说VLOOKUP函数的模糊匹配是怎么使用的？

如图6-57所示，这个一般用在区间的查找上，比如，根据区间查找等级。

在G2单元格中输入公式，并下拉填充公式。

=VLOOKUP(F2,A:C,3,1)

木木：哦，现在我懂了。

图 6-57　根据区间查找等级

6.5　如何借助文本函数进
行字符拆分与合并

《三国演义》第一回：“话说天下大势，分久必合，合久必分。周末七国分争，并入于

秦。及秦灭之后，楚、汉分争，又并入于汉。汉朝自高祖斩白蛇而起义，一统天下，后来光武

中兴，传至献帝，遂分为三国。”

其实Excel也经常干这种分分合合的事儿。

6.5.1　LEFT、MID、RIGHT函数提取部分字符

卢子：如图6-58所示，国有国法，群有群规，有很

多群进去都要重新更改备注名字，如我自

己：G-海珠-卢子，性别-地名-网名这样一

种形式。现在要如何将这些人员分成3列显

示，分别获取性别、地名、网名？

图 6-58　获取性别、地名、网名

木木：这样写备注挺好的，一眼就知道你在哪里工

作，是帅哥还是美女。如图6-59所示，如果

让我来做这个，直接用分列，分隔符号，

选中“其他”复选框，输入“-”即可。

图 6-59　按“-”分列

卢子：这个方法确实是最方便的，但是有一个缺

点就是当数据源更新时，不会自动更新，

得重新分列才可以，而这一点函数却能智

能办到。

性别就是左边1位，提取左边的函数用

LEFT，如图6-60所示，函数语法：

图 6-60　LEFT 函数语法

在B2单元格中输入公式，并下拉填充公式。

=LEFT(A2,1)

默认情况下，第二参数省略就是提取1
位，也可以这样写公式。

=LEFT(A2)

再看网名，这个是从右边提取，与LEFT
函数相反的就是RIGHT函数，如图6-61所
示，其语法与LEFT函数一样。

图 6-61　RIGHT 函数语法

木木：这样啊，那这个我来做。

在D2单元格中输入公式，并下拉填充 
公式。

=RIGHT(A2,2)

卢子：不错，就是这样。再说提取地名，也就是

提取中间的文本。如图6-62所示，用MID

函数，其语法如下。

图 6-62　MID 函数语法

这个函数与其他函数的差别在于多一个

参数——开始位置，也就是从哪一位开始提

取的。地名都是从第3位开始，提取2位，合

起来就是：

在C2单元格中输入公式，并下拉填充公式。

=MID(A2,3,2)

如图6-63所示，现在将“G-海珠-卢子”

改成“G-潮州-卢子”，效果立马更新，这是

技巧做不到的。

图 6-63　自动更新结果

木木：看来这几个函数还是有点用途的。

6.5.2　FIND函数辅助提取部分字符

卢子：之前的人员信息规律性非常强，一眼就看出

来。但现实中，很多人都是不统一的，就如

我的理财群一样，格式是“网名+职业”。网

名字符数不确定，有多有少，职业字符数也

不确定，如图6-64所示，在这种情况下又如

何提取呢？

图 6-64　提取网名与职业

木木：这么高难度的我不会。

卢子：虽然网名的字符数不确定，但其实还是有

规律的，就是在网名后面都是有分隔符号

“-”，也就是提取“-”前面1位就行。现

在的难点是如何确认这个“-”的位置？

如图6-65所示，查找文本在字符串中的

位置有一个专门的函数FIND，语法如下。

图 6-65　FIND 函数语法

=FIND("-",A2)

如图6-66所示，这样就可以轻松获取

“-”的位置。

图 6-66　获取“-”的位置

网名的字符数就是：

=FIND("-",A2)-1

提取左边的字符用LEFT函数，连起来 
就是：

=LEFT(A2,FIND("-",A2)-1)

职业的起始位置是“-”的位置+1位，也

就是：

=FIND("-",A2)+1

虽然职业的长度并不确定，但是职业在

最后面，只要提取的字符数大于职业的总长

度就可以提取到，也就是说，我们可以将提

取的长度写为4。
综合起来就是：

=MID(A2,FIND("-",A2)+1,4)

木木：怎么感觉在考数学题一样，有点晕晕的？

卢子：确实，不过还好，这些都是简单的四则运

算。如果不懂的话，可以先自己数一数，

多数几次就懂了。

6.5.3　使用“&”将内容合并起来

卢子：如图6-67所示，有分就有合，现在如何将拆

分的网名与职业合并起来呢？

图 6-67　合并网名与职业

木木：还真折腾，一下子分，一下子合。

卢子：学Excel就得折腾，才能学好。每次折腾

一下，就可以学到新的技能。如图6-68所

示，这里就要用到一个连字符“&”，语

法很简单。

图 6-68　“&”函数语法

这样就可以将内容合并起来：

=A2&B2

展知 识 扩

如果中间想加“-”，就可以用：

=A2&"-"&B2

“&”类似于月老，专门给人牵红线。要想将两个人合在一起，就用红绳绑住对方。

木木：月老是不是把我给忘了，怎么我现在都还没对象呢，呜呜。

卢子：是啊，老糊涂了，回头得好好提醒下他老人家才行。

当合并的单元格比较多的时候，用

“&”就会显得烦琐点，这时用PHONETIC
函数最适合，如图6-69所示。

图 6-69　合并多单元格内容

当然这个函数本身也有限制，只能连接

文本，数字却无法连接，这也是一个遗憾，

如图6-70所示。

图 6-70　无法连接数字

用合适的方法做合适的事，如果是多单元格的文本连接首选PHONETIC函数，如果包含数

字则用“&”连接。

6.6　小结和练习

函数与公式是Excel的精髓，可以通过函数与公式做很多你意想不到的数据处理方法，如

对数据进行查询、汇总、拆分、合并等。学好了我们就更进一个级别，而不仅仅停留在数据录

入员。

（1）如图6-71所示，这是一份产品销售明细表，现在要根据右边的价格对应表，获取单

价与金额。只要输入商品就自动显示单价，输入数量，就自动统计金额，部分结果如F、G两

列所示。

图 6-71　求单价与金额

（2）如图6-72所示，单价与金额出来后，就进行相应的汇总，在汇总表黄色区域输入公

式，汇总每个地区销售部门的金额。

图 6-72　汇总每个地区销售部门的金额


